
Agiformations Support de formation — Excel VBA

Avant-propos

Ce support accompagne la formation Excel VBA dispensée par A3iformations. Il a été concgu
comme un document de référence : vous pouvez le suivre pendant la session, puis le reprendre
seul pour consolider vos acquis et expérimenter a votre rythme.

La formation vise un objectif simple a énoncer mais riche en applications : connaitre les
principes généraux du VBA pour étendre les fonctionnalités d’Excel. Concrétement, vous
apprendrez a enregistrer puis écrire des macros, a manipuler les objets d’Excel (classeurs,
feuilles, plages), a créer des interfaces de saisie (formulaires), et a structurer un code robuste
capable d’automatiser vos taches récurrentes.

A qui s’adresse cette formation

Elle s’adresse aux salariés, indépendants et demandeurs d’emploi qui maitrisent déja les
fonctionnalités avancées d’Excel (formules, tableaux, références) et I'environnement Windows,
et qui souhaitent franchir le pas de I'automatisation. Aucune connaissance préalable en
programmation n’est exigée : les notions sont introduites progressivement.

Organisation du support
Le document suit fidelement le déroulé de la formation, en quatre séquences :
* Séquence 1 — Environnement et outils de débogage : découvrir I'éditeur VBE, lancer
et tracer une macro.

+ Séquence 2 — Les objets : variables, types, plages (Range), graphiques, énumérations,
modules de classe.
* Ségquence 3 — Les formulaires : boites de dialogue, UserForms et contrdles ActiveX.
+ Séquence 4 — Les procédures et fonctions avancées : procédures, conditions,
boucles, opérateurs, événements et ouverture vers ADO/DAO et 'API Windows.
Chaque séquence comprend des explications, des exemples de code commentés, des pas-a-
pas, des points de vigilance et un tableau récapitulatif. Des exercices gradués (Découverte,
Application, Maitrise) cl6turent chaque séquence ; leurs corrigés sont regroupés en fin de
document, suivis d’un atelier fil rouge et d’annexes pratiques.

Bon a savoir
Formation éligible au CPF et finan¢able dans le cadre du plan de développement des compétences.
Une attestation de fin de formation est remise a l'issue de la session.

Conventions typographiques

Le code VBA est présenté en police a chasse fixe sur fond gris. Les noms de menus, boutons
et touches apparaissent en gras. Trois pictogrammes ponctuent le texte :

Astuce
Astuce — un raccourci, une bonne pratique ou un gain de temps.

Attention
Attention — une erreur fréquente ou un piege a éviter.

Bon a savoir
Bon a savoir — un complément utile pour aller plus loin.

Séquence 1 — Environnement et outils de débogage

Objectifs pédagogiques

Connaitre et personnaliser I'environnement logiciel
Reconnaitre l'interface de I'éditeur Visual Basic (VBE)
Lancer, arréter et tracer I'exécution d’'une macro
Connaitre I'environnement et les outils de débogage

Prérequis : Connaitre parfaitement Excel et I'environnement Windows.

1.1 Afficher I'onglet Développeur

Par défaut, 'onglet Développeur est masqué dans le ruban d’Excel. C’est pourtant le point
d’entrée vers le VBA : il donne accés a I'éditeur Visual Basic, a I'enregistreur de macros et aux
contrbles de formulaire.

Pas a pas — activer 'onglet :

1. Ouvrez le menu Fichier puis Options.
2. Cliquez sur Personnaliser le ruban.
3. Dans la colonne de droite, cochez Développeur, puis validez par OK.

Classeurl - Excel X 0O _
Fichier Accueil Insertion Mise en page Formules Données Révision Affichage Développeur
Visual Basic Enregistrer Insérer Sources

Visual Basic / Macros Enregistrer / une macra Insérer [ Mode Création Sources [ XML
Onglet Développeur : porte d'entrée vers Visual Basic, l'enregistreur de macros et les contriles.

L’onglet Développeur une fois activé dans le ruban Excel.

Astuce
Raccourci direct vers I'éditeur depuis Excel : A1t + F11. La méme combinaison fait revenir a la
feuille de calcul.

1.2 Enregistrer puis lancer une premiere macro

L’enregistreur de macros traduit vos actions en code VBA. C’est un excellent point de départ
pour découvrir la syntaxe : on agit dans Excel, on lit le code généré, puis on I'adapte.

Pas a pas — enregistrer une macro :

4. Onglet Développeur > Enregistrer une macro.

5. Donnez un nom (sans espace ni accent), éventuellement une touche de raccourci, puis
OK.

6. Réalisez quelques actions (mise en forme, saisie...).
7. Cliquez sur Arréter I’enregistrement.

22 Route d’Abbeville — 62390 Auxi-le-Chéateau Page 6/ 32 contact@ag3iformations.fr



A3iformations Support de formation — Excel VBA

Pour relancer la macro : Développeur > Macros (ou Alt + F8), sélectionnez-la et cliquez sur
Exécuter. Vous pouvez aussi lui associer un bouton ou une icéne du ruban.

Bon a savoir

Un classeur contenant des macros doit étre enregistré au format .xlsm (classeur prenant en charge
les macros). Au format .xIsx, le code est perdu a la fermeture.

1.3 Reconnaitre I’'interface de I’éditeur VBA (VBE)

L’éditeur Visual Basic — le VBE — est une application distincte d’Excel. Il s’organise autour de
trois fenétres principales :

* L’explorateur de projets (en haut a gauche) : il liste les objets Excel (feuilles,
ThisWorkbook), les modules standards et les modules de classe.

+ Lafenétre Propriétés (en bas a gauche) : elle affiche et modifie les propriétés de
I'élément sélectionné.

+ Lafenétre de code (a droite) : c’est la que I'on écrit et lit les macros.

Microsoft Visual Basic for Applications - Classeurl

Fchier Edition Affichage Insertion Format Debogage Exécution Outils Fenétre ?
Projet - VBAProject (General) Bonjour

B VBAProject (Classeurl)
HE Microsoft Excel Objets
Feuill {Feuill)

Option Explicit

Feuil2 (Feuil2) Sub Eon]our{} .
ThisWorkbaook Dim  prenom As 5tring
B Modules prenom = InputBox ("Votre prénom ?" )
Modulel MsqBox  "Bonjour " & prenom, vbInformation
B Modules de classe End Sub
Clculs
Propriétés - Modulel ' Commentaire : explique le rdle de la macro
(Name) Modulel

L'editeur VBE regroupe l'explorateur de projets, la fenetre Proprietes et la fenétre de code.

L’éditeur VBE : explorateur de projets, fenétre Propriétés et fenétre de code.

La barre de menus donne accés aux commandes (Insertion, Débogage, Exécution...) et la
barre d’outils aux actions courantes (Exécuter, Pas a pas, Arrét).

1.4 Les modules et la structure d’une macro

Le code se range dans des modules. On distingue les modules standards (code partagé par
tout le classeur), les modules d’objet (ThisWorkbook et chaque feuille, qui regoivent les
procédures d’événements) et les modules de classe (vus en séquence 2).

Pour créer un module standard : menu Insertion > Module. Une macro y prend la forme d’'une
procédure Sub :

22 Route d’Abbeville — 62390 Auxi-le-Chéateau Page 7/ 32 contact@ag3iformations.fr



Agiformations Support de formation — Excel VBA

Sub PremiereMacro()
' Les commentaires commencent par une apostrophe

Range("A1").Value = "Bonjour" ' on écrit dans la cellule Al
Range("A1").Font.Bold = True ' on met le texte en gras
End Sub

Une procédure commence par Sub suivi d’'un nom et de parenthéses, et se termine par End
Sub. Les commentaires, introduits par une apostrophe, sont ignorés a I'exécution mais
essentiels a la lisibilité.

Astuce

Commentez systématiquement le réle d’'une macro en premiére ligne. Six mois plus tard, c’est vous
qui vous remercierez.

1.5 Tracer et déboguer un code

Déboguer, c’est exécuter un code pas a pas pour comprendre son comportement et localiser
une erreur. Le VBE offre pour cela plusieurs outils.

Exécution en mode Pas a pas

La touche F8 exécute le code une instruction a la fois. La ligne sur le point d’étre exécutée est
surlignée en jaune : on suit ainsi le cheminement réel du programme, y compris dans les
conditions et les boucles.

Modulel (Code) - Pas a pas détaillé (F8)

Sub CalculRemise()
Dim prix As Double, remise As Double
prix = Range("B2").Value
- If prix > 188 Then
remise = prix * 0.1
Else
remise = @
End If
Range("C2").Value = prix - remise
End Sub

Exécution Pas a pas (F8) : la ligne courante est surlignée en jaune.

Les fenétres de surveillance

* Variables locales (Affichage > Fenétre Variables locales) : affiche en temps réel la valeur
et le type de toutes les variables de la procédure.

« Espions (Débogage > Ajouter un espion) : surveille une expression précise et peut
suspendre I'exécution dés qu’elle change ou devient vraie.

+ Exécution (immédiate) (Ctrl + G) : teste une instruction a la volée. ? maVariable affiche
sa valeur ; Debug.Print y écrit pendant I'exécution.

| Espions | Exécution |
Expression Valeur Type Expression Valeur Type 7 total

prenom "Manu" String total > 10680 Vrai Boolean 1540

total 1540 Long Cells(1,1) Date" Variant 7 UCase(prenom)

taux 8,2 Double MANU

ok Vrai Boolean Debug.Print Now

16/06/2026 09:14

Fenétres de debogage : Variables locales, Espions et Execution immediate (7 = Print).
Variables locales, Espions et fenétre Exécution immédiate.

Points d’arrét et arrét d’'une macro
Un clic dans la marge grise (ou F9) place un point d’arrét : 'exécution s’interrompt sur cette
ligne. Pour arréter une macro en cours, utilisez Echap ou Ctrl + Pause.

Messages d’erreur et explorateur d’objets

Quand une instruction échoue, VBA affiche une boite d’erreur avec un numéro et un libellé ; le
bouton Débogage surligne la ligne fautive. L’Explorateur d’objets (F2) permet quant a lui de
parcourir toutes les classes, propriétés et méthodes disponibles.

Explorateur d'objets X [ _
| Excel v |
Classes Membres de 'Range’
Application W Value Propriete
Workbook W Cells t
Worksheet W Offset
Range W Font
Chart W Interior Proprigte
ChartObject # Select Methode
Font + Copy Methode
Interior # ClearContents Methode
# Sort Methode

L'Explorateur d'objets (F2) liste les classes, proprietés (W) et methodes (¢) de chaque bibliotheque.

L’Explorateur d’objets (F2) : classes, propriétés (m) et méthodes ( #).

1.6 Personnaliser et sécuriser I’environnement

- Forcer la déclaration des variables : Outils > Options > onglet Editeur > « Déclaration
des variables obligatoire ». Chaque nouveau module recoit alors l'instruction Option
Explicit.

» Créer uneicbne sur le ruban : Fichier > Options > Personnaliser le ruban pour ajouter
un bouton langant votre macro.

* Lancement automatique : une macro nommée Auto_Open ou la procédure d’événement
Workbook_Open s’exécute a I'ouverture du classeur.

« Seécuriser le code par mot de passe : Outils > Propriétés de VBAProject > onglet
Protection.

Attention

Le mot de passe d’'un projet VBA protege I'affichage du code, pas vos données. Notez-le
précieusement : un projet verrouillé sans mot de passe est trés difficile a rouvrir.

Tableau récapitulatif — Séquence 1

Action Commande / raccourci

Ouvrir I'éditeur VBE Alt + F11
Lancer la liste des macros Alt + F8
Exécuter la procédure courante F5

Exécution pas a pas détaillé F8

Point d’arrét sur une ligne F9

Fenétre Exécution immédiate Ctrl + G
Explorateur d’objets F2

Forcer la déclaration des variables Option Explicit

Exercices — Séquence 1
Exercice 1.1 Premiére prise en main
* Affichez 'onglet Développeur, puis ouvrez I'éditeur VBE avec Alt + F11.
* Insérez un module standard et recopiez la procédure PremiereMacro de la section 1.4.
* Exécutez-la avec F5, puis recommencez en mode pas a pas avec F8.
Exercice 1.2 Tracer une exécution
* Ajoutez a la macro une ligne qui écrit la date du jour en A2 (indice : Date).
* Placez un point d’arrét sur cette ligne (F9) et lancez la macro.
* Dans la fenétre Exécution, tapez ? Date puis Entrée et observez le résultat.
Exercice 1.3 Automatiser a 'ouverture

* Dans le module ThisWorkbook, écrivez une procédure Workbook Open qui affiche un message
d’accueil au démarrage du classeur.

* Enregistrez le classeur au format .xIsm, fermez-le puis rouvrez-le pour vérifier.

Séquence 2 — Les objets

Objectifs pédagogiques

Créer et utiliser des modules de classe

Connaitre les objets et les types de variables

Savoir créer l'instruction Enum et manipuler 'objet Range
Editer des graphiques par le code

Prérequis : Connaitre les principes généraux du VBA (séquence 1).

2.1 Le modeéle objet d’Excel

VBA pilote Excel a travers un modeéle objet hiérarchique. L’objet Application (Excel lui-méme)
contient des classeurs (Workbooks), qui contiennent des feuilles (Worksheets), qui contiennent
des plages (Range). On descend la hiérarchie avec le point :

Application.Workbooks("Budget.xlsm").Worksheets("Janvier").Range("Al").Value = 100

Raccourcis usuels :

Range("A1") ' plage de la feuille active

Cells(1, 1) ' cellule ligne 1, colonne 1 (= Al)
ActiveSheet.Name ' nom de la feuille active
ThisWorkbook.Save ' enregistre le classeur contenant le code

Chaque objet posséde des propriétés (ses caractéristiqgues, comme Value ou Font) et des
méthodes (les actions qu’il sait faire, comme Copy ou Select). L’Explorateur d’objets (F2) en
donne la liste exhaustive.

2.2 Les variables et leurs types

Une variable est un nom qui désigne un emplacement mémoire dans lequel on range une
valeur. On la déclare avec Dim, en précisant son type :

Dim prenom As String ' texte
Dim age As Integer ' entier (-32 768 a 32 767)
Dim total As Long ' grand entier

nombre décimal

Vrai / Faux

date et heure

objet (plage de cellules)
n'importe quel type

Dim prix As Double
Dim livre As Boolean
Dim jour As Date

Dim cellule As Range
Dim divers As Variant

Le tableau ci-dessous résume les types les plus courants :

Type ' Contenu Exemple de valeur
String Texte "Auxi-le-Chateau"
Integer Entier court 42

Long Entier long 1500000

Type ' Contenu Exemple de valeur
Double Décimal 19,99

Boolean Booléen True / False

Date Date / heure #06/16/2026#
Variant Type indéterminé tout type

Forcer et limiter la portée

Placée en téte de module, l'instruction Option Explicit oblige & déclarer chaque variable :
c’est la meilleure protection contre les fautes de frappe. La portée d’une variable dépend de
'endroit ou on la déclare :

* Dans une procédure (Dim) : la variable n’existe que le temps de la procédure (portée
locale).

* En téte de module (Dim ou Private) : la variable est partagée par toutes les procédures
du module.

* Avec Public en téte de module : la variable est accessible depuis tout le projet.
Attention

Sans Option Explicit, une variable mal orthographiée (ex. totl au lieu de total) est créée
silencieusement a vide : le résultat est faux sans aucun message d’erreur.

2.3 Les constantes et les conversions de type

Une constante est une valeur fixe nommée, déclarée avec Const. VBA fournit aussi des
constantes intégrées (vbinformation, vbCrLf, xIUp...).

Const TVA As Double = 0.2
Const SOCIETE As String = "A3iformations”

Fonctions de conversion :

montant = CCur("1540,50") ' texte -> monétaire (Currency)
d = CDate("16/06/2026") ' texte -> date

ok = IsDate("31/02/2026") ' False : date invalide

n = CInt("42") ' texte -> entier

Bon a savoir
IsDate et IsNumeric vérifient qu'une saisie est convertible avant de tenter la conversion :
indispensable pour fiabiliser les entrées d’'un formulaire.

2.4 L'’objet Range et I'instruction Set

Pour ranger un objet dans une variable, on emploie Set (et non un simple =, réservé aux
valeurs). On peut alors enchainer les manipulations sans répéter le chemin complet :

Dim plage As Range
Set plage = Worksheets("Ventes").Range("B2:B100")

plage.Value = © vide la plage

plage.Font.Bold = True ' met en gras
plage.Interior.Color = vbYellow ' fond jaune
MsgBox plage.Cells.Count & " cellules"

' Se déplacer avec Offset et trouver la derniére ligne :
Dim derniere As Long
derniere = Cells(Rows.Count, 1).End(x1Up).Row

Astuce
Le bloc With .. End With évite de répéter le nom de I'objet et accélere le code :

With Range("A1")
.Value = "Total"
.Font.Bold = True
.Interior.Color = vbCyan
End With

2.5 Les tableaux

Un tableau (array) stocke plusieurs valeurs dans une méme variable, repérées par un indice. Il
est tres efficace pour traiter de gros volumes en mémoire plutbt que cellule par cellule.

Dim mois(1 To 12) As String ' tableau de 12 éléments
mois(1) = "Janvier"
mois(2) = "Février"

Charger une plage entiere en mémoire (trés rapide) :

Dim donnees As Variant

donnees = Range("A1:C100").Value ' tableau a 2 dimensions
MsgBox donnees(5, 2) ' ligne 5, colonne 2

2.6 Créer une énumération (instruction Enum)

Une énumération définit une liste de constantes nommeées et lisibles, a utiliser en place de
valeurs numériques obscures.

Enum NiveauFormation
Basique = 1
Operationnel = 2
Avance = 3

End Enum

Dim n As NiveauFormation
n = Operationnel
If n = Avance Then MsgBox "Niveau expert"

Bon a savoir
Une énumeération propose I'auto-complétion : en tapant le nom du type, VBA affiche la liste des
valeurs possibles. Le code devient auto-documente.

2.7 Manipuler et éditer des graphiques

VBA crée et met en forme des graphiques par le code, ce qui est précieux pour générer
automatiquement des tableaux de bord.

Dim graph As ChartObject
Set graph = ActiveSheet.ChartObjects.Add(Left:=200, Top:=20, _
Width:=360, Height:=220)
With graph.Chart
.SetSourceData Source:=Range("A1:B12")
.ChartType = x1ColumnClustered
.HasTitle = True
.ChartTitle.Text = "Ventes mensuelles"
End With

Astuce
Le caractere _ (espace puis souligné) en fin de ligne permet de poursuivre une instruction sur la ligne
suivante, pour garder un code lisible.

2.8 Les modules de classe

Un module de classe permet de définir vos propres objets, avec leurs propriétés et méthodes.
C’est une notion avancée, utile pour structurer de grosses applications.

Dans un module de classe nommé "Stagiaire"
Public Nom As String
Public Niveau As NiveauFormation

Public Function EstAvance() As Boolean
EstAvance = (Niveau = Avance)

End Function

' Utilisation dans un module standard :

Dim s As New Stagiaire

s.Nom = "Manu" : s.Niveau = Avance

MsgBox s.Nom & " - expert : " & s.EstAvance()

Bon a savoir
Les feuilles de calcul et ThisWorkbook sont elles-mémes des modules de classe particuliers : c’est
pourquoi elles possédent des événements (Worksheet Change, etc.).

Tableau récapitulatif — Séquence 2

Notion | Mot-clé / syntaxe Role

Déclarer une variable Dim x As Type Réserver un emplacement typé
Affecter un objet Set p = Range(...) Lier une variable objet
Constante Const TVA = 0.2 Valeur fixe nommée
Enumération Enum .. End Enum Liste de constantes lisibles
Factoriser With .. End With Eviter de répéter 'objet
Tableau Dim t(1 To 12) Stocker plusieurs valeurs

Exercices — Séquence 2
Exercice 2.1 [Mal Variables et Range
* Déclarez une variable taux (Double) a 0,2 et une variable ville (String).
 Ecrivez en A1 le texte « TVA » et en B1 la valeur de taux, a 'aide de Range.
Exercice 2.2 Parcourir et mettre en forme
e Trouvez la derniére ligne remplie de la colonne A (End(xlUp).Row).
* Avec un bloc With, mettez la plage Al:A[derniere] en gras et fond jaune.
Exercice 2.3 Enumération et graphique
e Créez une énumération NiveauFormation (Basique/Opérationnel/Avancé).
A partir d’une plage A1:B12, générez par le code un graphique en colonnes intitulé « Synthése ».

Séquence 3 — Les formulaires

Objectifs pédagogiques

Identifier une boite de dialogue et un formulaire

Insérer un contréle ActiveX et créer un UserForm

Afficher des messages et récupérer les réponses de I'utilisateur
Modifier les propriétés des controles

Prérequis : Connaitre l'interface, les objets et les procédures de base.

3.1 Les boites de dialogue intégrées : MsgBox et InputBox

Avant de créer ses propres formulaires, VBA fournit deux boites de dialogue prétes a 'emploi :
MsgBox pour informer ou interroger, InputBox pour recueillir une saisie.

MsgBox — afficher un message

MsgBox "Traitement terminé avec succés.", vbInformation, "A3iformations"
' Poser une question et récupérer la réponse :
Dim rep As VbMsgBoxResult

rep = MsgBox("Enregistrer les modifications ?", vbYesNo + vbQuestion)
If rep = vbYes Then ThisWorkbook.Save

Microsoft Excel X0 _

o Traitement terminé avec succes.
12 lignes ont été mises a jour.
oK

MsgBox "Traitement terminé...", vbInformation, "Microsoft Excel"

Une boite MsgBox de type information avec un bouton OK.

Le deuxiéme argument combine une icbne (vbinformation, vbExclamation, vbQuestion,
vbCritical) et des boutons (vbOKOnly, vbYesNo, vbYesNoCancel...). La fonction renvoie le
bouton cliqué (vbYes, vbNo, vbCancel...).

InputBox — recueillir une saisie
Dim taux As String

taux = InputBox("Entrez le taux de TVA (en %) :", "Saisie", "20")
If taux <> "" Then Range("D1").Value = CDbl(taux) / 100

Saisie xO_
Entrez le taux de TVA (en %) : oK
Annuler
20
taux = InputBox("Entrez le taux de TVA (en %) :", "Saisie")

Une boite InputBox avec valeur par défaut proposée.

Attention
InputBox renvoie toujours du texte. Pensez a convertir (CDbl, Cint...) et a vérifier le résultat : si
I'utilisateur clique sur Annuler, la fonction renvoie une chaine vide.

3.2 Les contrbles ActiveX dans la feuille

Un contréle ActiveX (bouton, case a cocher, liste déroulante...) s'insére directement dans une
feuille de calcul pour créer une mini-interface.

Pas a pas — insérer un bouton de commande :

8. Onglet Développeur > Insérer > zone « Contrdles ActiveX » > Bouton de commande.
9. Dessinez le bouton sur la feuille en maintenant le clic.

10. Double-cliquez dessus (en Mode Création) : le VBE s’ouvre sur la procédure
d’événement Click.

11. Saisissez le code, puis désactivez le Mode Création pour utiliser le bouton.

Private Sub CommandButtonl_Click()
MsgBox "Bouton cliqué !"
End Sub

3.3 Créer un UserForm

Le UserForm est un formulaire personnalisé, beaucoup plus riche qu’une simple boite de
dialogue. Il regroupe des contrdles (zones de texte, listes, boutons) pour une saisie structurée.
Pas a pas — créer un formulaire :

12. Dans le VBE : menu Insertion > UserForm.
13. Depuis la Boite a outils, déposez les contréles voulus sur le formulaire.

14. Sélectionnez chaque contrdle et réglez ses propriétés dans la fenétre Propriétés
(notamment Name et Caption).

15. Double-cliquez sur un bouton pour écrire le code qui s’exécute au clic.

Boite a outils

Fiche client

Nom :

CommandButton Prénom -
Frame:

Socité -

Prafil -

5 la Boite & outils

Un UserForm « Fiche client » en mode Création, avec sa Boite a outils.

Les principaux contrdles

Controle Usage Propriété clé
Label Texte fixe (libellé) Caption
TextBox Zone de saisie Text
ComboBox Liste déroulante List

ListBox Liste de choix ListIndex
CheckBox Case a cocher Value
OptionButton Bouton radio Value
CommandButton Bouton d’action Caption

Faire vivre le formulaire

On affiche le formulaire avec sa méthode Show, on lit ses contréles, puis on le ferme avec
Unload Me. Voici un exemple complet de bouton « Valider » :

Affichage depuis un module standard :
Sub OuvrirFiche()

UserForml.Show
End Sub
' Code du bouton Valider, dans le UserForm :
Private Sub btnValider_Click()

If txtNom.Text = "" Then
MsgBox "Le nom est obligatoire.", vbExclamation
Exit Sub

End If

Dim L As Long

L = Cells(Rows.Count, 1).End(x1lUp).Row + 1
Cells(L, 1).Value = txtNom.Text

Cells(L, 2).Value = txtPrenom.Text

Cells(L, 3).Value = cboProfil.Value
Unload Me ' ferme le formulaire
End Sub

Astuce
Pré-remplissez une liste déroulante a 'ouverture du formulaire dans I'événement
UserForm_Initialize, grace a la méthode AddItem: cboProfil.AddItem "Salarié".

3.4 L’ordre des tabulations

L’ordre dans lequel la touche Tab fait passer d’'un champ a l'autre se régle via Affichage >
Ordre de tabulation (ou la propriété TabIndex de chaque contrdle). Un ordre cohérent est
essentiel au confort de saisie.

Tableau récapitulatif — Séquence 3

Besoin Outil VBA

Informer l'utilisateur MsgBox "..", vbInformation
Poser une question oui/non MsgBox("..", vbYesNo)
Demander une saisie simple InputBox("..")

Interface de saisie compléte UserForm + controles

Afficher / fermer un UserForm Show / Unload Me

Remplir une liste ComboBox.AddItem

Exercices — Séquence 3
Exercice 3.1 Dialogue avec l'utilisateur
* Affichez un message d’'information « Bienvenue » avec une icbne.
¢ Demandez le prénom via InputBox et affichez « Bonjour [prénom] ».
Exercice 3.2 Un bouton qui agit
* Insérez un bouton ActiveX dans une feuille.
* Auclic, il efface la plage A1:C100 apres confirmation par vbYesNo.
Exercice 3.3 Mini fiche de saisie
e Créez un UserForm avec Nom, Prénom, une liste déroulante Profil et un bouton Valider.

* Pré-remplissez la liste a 'ouverture, contrélez que le Nom est saisi, puis écrivez la ligne a la suite
des données existantes.

Séquence 4 — Les procédures et fonctions avanceées

Objectifs pédagogiques

Savoir créer et modifier des procédures (Sub et Function)
Maitriser les conditions, boucles et opérateurs

Comprendre les événements et I'appel entre macros

Découvrir le pilotage d’autres applications et I'accés aux données

Prérequis : Connaitre l'interface, les objets, les procédures de base et les formulaires.

4.1 Sub, Function et arguments

Une procédure Sub exécute des actions sans renvoyer de valeur. Une procédure Function
calcule et renvoie un résultat — on peut méme l'utiliser comme une formule personnalisée dans
une cellule.

Procédure avec arguments :

Sub Saluer(prenom As String)
MsgBox "Bonjour " & prenom

End Sub

' Fonction personnalisée, utilisable dans une cellule : =TTC(100)

Function TTC(prixHT As Double, Optional taux As Double = ©0.2) As Double
TTC = prixHT * (1 + taux)

End Function

Le mot-clé Optional rend un argument facultatif (ici, le taux vaut 0,2 par défaut). On appelle
une macro depuis une autre simplement par son nom, ou avec Call : Call Saluer("Manu").

Bon a savoir
Une fonction personnalisée (UDF) apparait dans I'assistant fonction d’Excel, catégorie «
Personnalisées ». Idéal pour packager un calcul métier réutilisable.

4.2 Activer, sélectionner... ou s’en passer

L’enregistreur de macros produit beaucoup d’instructions Select et Activate. Or il est presque
toujours préférable d’agir directement sur les objets, sans les sélectionner : le code est plus
rapide et plus fiable.

A éviter (lent, fragile) :
Range("A1").Select
Selection.Value = 100

' A préférer (direct) :
Range("A1").Value = 100

4.3 Les opérateurs

Catégorie Opérateurs | Exemple
Arithmétiques + - * / \ Mod ~ 17 Mod 5 = 2
Comparaison = <> < > <K= >= prix >= 100
Concaténation & "A" & "B" = "AB"
Logiques And Or Not Xor a>0And b >0

4.4 Les conditions : If et Select Case

If note >»>= 16 Then

mention = "Trés bien"
ElseIf note >= 12 Then

mention = "Bien"
Else

mention = "A revoir"
End If

Plus lisible quand on teste une méme variable :
Select Case niveau

Case 1: libelle = "Basique"

Case 2: libelle = "Opérationnel"

Case Else: libelle = "Avancé"
End Select

4.5 Les boucles
Une boucle répéte un bloc d’instructions. VBA en propose plusieurs formes selon le besoin :

Répéter un nombre connu de fois :
For i =1 To 10

Cells(i, 1).Value = i * i
Next i
' Parcourir une collection d'objets :
For Each ¢ In Range("A1:A20")

If c.Value < @ Then c.Interior.Color = vbRed
Next c

Répéter tant qu'une condition est vraie :
Do While Cells(L, 1).Value <> ""
L=L+1
Loop

Attention
Une boucle Do While dont la condition ne devient jamais fausse tourne a I'infini. En cas de blocage,
arrétez avec Echap ou Ctrl + Pause.

4.6 Les événements

Un événement est une procédure déclenchée automatiquement par une action de I'utilisateur
ou du systeme. On les écrit dans le module de la feuille ou de ThisWorkbook.

Dans le module d'une feuille : réagir a chaque modification
Private Sub Worksheet_Change(ByVal Target As Range)
If Not Intersect(Target, Range("B:B")) Is Nothing Then

Target.Offset(0, 1).Value = Now ' horodate en colonne C
End If
End Sub

Evénement Déclenché quand...
Workbook_Open le classeur s’ouvre
Workbook_BeforeClose avant la fermeture du classeur
Worksheet_Change une cellule de la feuille est modifiée
Worksheet_SelectionChange la sélection change sur la feuille

4.7 Gérer les erreurs d’exécution

Un code robuste anticipe les erreurs (fichier introuvable, division par zéro...) plutét que
d’afficher un message brut a I'utilisateur.

Sub TraitementSur()

On Error GoTo Gestion

' .. instructions susceptibles d'échouer ..

Exit Sub
Gestion:

MsgBox "Erreur " & Err.Number & " : " & Err.Description, vbCritical
End Sub

Astuce
On Error Resume Next ignore I'erreur et passe a la ligne suivante : pratique ponctuellement, mais a
manier avec prudence pour ne pas masquer de vrais problemes.

4.8 Fonctions avancées et ouverture

La séquence se conclut par un panorama des possibilités avancées, qui font du VBA un
véritable langage de développement :

+ Gérer les compléments : regrouper du code réutilisable dans un complément (.xlam)
chargé dans tous vos classeurs.

+ Piloter les autres applications Office (OLE) : depuis Excel, créer un message Outlook,
alimenter un document Word ou une présentation PowerPoint.

* Accéder a des données avec ADO et DAO : interroger une base Access ou un serveur
SQL directement depuis le code.

» Faire appel aux fonctions API de Windows : via l'instruction Declare, utiliser des
fonctions systeme non exposées par VBA (notion experte).

' Exemple OLE : envoyer un mail via Outlook
Dim ol As Object, msg As Object

Set ol = CreateObject("Outlook.Application")
Set msg = ol.CreateItem(9)

With msg

.To = "stagiaire@exemple.fr"

.Subject = "Convocation formation VBA"

.Body = "Bonjour, votre session débute lundi."

.Display ' .Send pour envoyer directement
End With

Bon a savoir
CreateObject (liaison tardive) évite d’ajouter une référence : le code reste portable d’'un poste a
I'autre, méme si les versions d’'Office different.

Tableau récapitulatif — Séquence 4

Notion Syntaxe clé

Procédure / fonction Sub .. End Sub / Function .. End Function
Argument facultatif Optional taux As Double = 0.2

Condition If .. ElseIf .. Else / Select Case
Boucles For..Next / For Each / Do While..Loop
Evénement Private Sub Worksheet_Change(...)
Gestion d’erreur On Error GoTo .. / Err.Number

Piloter Office CreateObject("Outlook.Application")

Exercices — Séquence 4
Exercice 4.1 Premiére fonction
* Ecrivez une fonction TTC(prixHT) qui renvoie le prix TTC (taux 20 %).
¢ Testez-la dans une cellule : =TTC(100) doit afficher 120.
Exercice 4.2 Boucle et condition
¢ Parcourez la plage A1:A20 avec For Each.
* Colorez en rouge les valeurs négatives, en vert celles supérieures a 1000.
Exercice 4.3 Evénement et robustesse
* Dans Worksheet_Change, horodatez en colonne C toute saisie faite en colonne B.
* Protégez le code par On Error pour éviter tout message brut en cas d’erreur.

Corrigés des exercices

Les corrigés proposent une solution parmi d’autres : en programmation, plusieurs écritures
correctes coexistent souvent. L’essentiel est que le code soit lisible, juste et robuste.

Corrigé 1.1 — Premiére prise en main
Il suffit de recopier la procédure et de I'exécuter. En mode pas a pas (F8), on observe que
chaque ligne s’exécute dans I'ordre, de Sub a End Sub.

Corrigé 1.2 — Tracer une exécution

Sub PremiereMacro()
Range("A1").Value = "Bonjour"
Range("A2").Value = Date

End Sub

Dans la fenétre Exécution, ? Date affiche la date du jour au format systeme.
Corrigé 1.3 — Automatiser a I'ouverture

' Module ThisWorkbook
Private Sub Workbook_Open()

MsgBox "Bienvenue dans le classeur de formation VBA !"
End Sub

Corrigé 2.1 — Variables et Range

Dim taux As Double, ville As String
taux = 0.2

Range("A1").Value
Range("B1").Value

n TVAII
taux

Corrigé 2.2 — Parcourir et mettre en forme

Dim derniere As Long
derniere = Cells(Rows.Count, 1).End(x1lUp).Row
With Range("Al:A" & derniere)
.Font.Bold = True
.Interior.Color = vbYellow
End With

Corrigé 2.3 — Enumération et graphique

Enum NiveauFormation
Basique = 1
Operationnel = 2
Avance = 3

End Enum

Sub Synthese()
Dim g As ChartObject
Set g = ActiveSheet.ChartObjects.Add(200, 20, 360, 220)
With g.Chart
.SetSourceData Range("A1l:B12")
.ChartType = x1ColumnClustered
.HasTitle = True : .ChartTitle.Text = "Synthese"

End With
End Sub

Corrigé 3.1 — Dialogue avec l'utilisateur

MsgBox "Bienvenue", vbInformation
Dim p As String

p = InputBox("Votre prénom ?")
MsgBox "Bonjour " & p

Corrigé 3.2 — Un bouton qui agit

Private Sub CommandButtonl_ Click()
If MsgBox("Effacer A1:C100 ?", vbYesNo + vbQuestion) = vbYes Then
Range("A1:C100").ClearContents
End If
End Sub

Corrigé 3.3 — Mini fiche de saisie

Private Sub UserForm_Initialize()
cboProfil.AddItem "Salarié"
cboProfil.AddItem "Indépendant"
cboProfil.AddItem "Demandeur d'emploi”

End Sub

Private Sub btnValider_Click()
If txtNom.Text = "" Then MsgBox "Nom obligatoire": Exit Sub
Dim L As Long
L = Cells(Rows.Count, 1).End(xlUp).Row + 1
Cells(L, 1) = txtNom.Text
Cells(L, 2) = txtPrenom.Text
Cells(L, 3) = cboProfil.Value
Unload Me
End Sub

Corrigé 4.1 — Premiére fonction

Function TTC(prixHT As Double, Optional taux As Double = ©0.2) As Double
TTC = prixHT * (1 + taux)
End Function

Corrigé 4.2 — Boucle et condition

Dim c As Range
For Each ¢ In Range("Al:A20")
If c.Value < @ Then
c.Interior.Color = vbRed
ElseIf c.Value > 1000 Then
c.Interior.Color = vbGreen
End If
Next c

Corrigé 4.3 — Evénement et robustesse

Private Sub Worksheet_Change(ByVal Target As Range)

On Error Resume Next

If Not Intersect(Target, Range("B:B")) Is Nothing Then
Target.Offset(0, 1).Value = Now

End If
End Sub

Atelier fil rouge — Mini-application « Suivi des stagiaires »

Cet atelier de synthese mobilise les notions des quatre séquences pour construire une petite
application compléte : un classeur qui enregistre des stagiaires via un formulaire, génére un
tableau de bord et envoie une convocation par mail.

Bon a savoir

Réalisez I'atelier par étapes, en testant chaque brique avant de passer a la suivante. C’est la
meilleure fagcon de déboguer sereinement.

Etape 1 — Préparer le classeur

* Créez une feuille « Données » avec, en ligne 1, les en-tétes : Nom, Prénom, Société,
Niveau, Date.

» Définissez une énumération NiveauFormation (Basique, Opérationnel, Avancé).
+ Enregistrez le classeur au format .xIsm.

Etape 2 — Le formulaire de saisie

* Créez un UserForm « frmStagiaire » avec : txtNom, txtPrenom, txtSociete, cboNiveau et
deux boutons (Valider, Fermer).

* Dans UserForm_Initialize, remplissez cbhoNiveau avec les trois niveaux.

* Au clic sur Valider : contrdlez que le Nom est saisi, puis ajoutez une ligne dans « Données
» avec la date du jour (Date).

Private Sub btnValider_Click()

If txtNom.Text = "" Then MsgBox "Le nom est obligatoire.", vbExclamation: Exit Sub

Dim ws As Worksheet, L As Long

Set ws = Worksheets("Données")

L = ws.Cells(ws.Rows.Count, 1).End(xlUp).Row + 1

ws.Cells(L, 1) = txtNom.Text

ws.Cells(L, 2) = txtPrenom.Text

ws.Cells(L, 3) = txtSociete.Text

ws.Cells(L, 4) = cboNiveau.Value

ws.Cells(L, 5) = Date

MsgBox "Stagiaire enregistré.", vbInformation

txtNom = "": txtPrenom = "": txtSociete =
End Sub

réinitialise

Etape 3 — Le tableau de bord

+ Sur une feuille « Tableau de bord », écrivez une macro qui compte le nombre de
stagiaires par niveau (boucle ou NB.SI piloté par le code).

« Générez par le code un graphique en colonnes représentant cette répartition.

Sub MajTableauBord()
Dim ws As Worksheet, td As Worksheet, derniere As Long, c As Range
Dim nB As Long, nO As Long, nA As Long
Set ws = Worksheets("Données")
derniere = ws.Cells(ws.Rows.Count, 1).End(x1Up).Row

For Each ¢ In ws.Range("D2:D" & derniere)
Select Case c.Value
Case "Basique": nB = nB + 1
Case "Opérationnel”: nO = nO + 1
Case "Avancé": nA = nA + 1
End Select
Next c
Set td = Worksheets("Tableau de bord")
td.Range("Al1:B4").Value = Array( _
Array("Niveau", "Effectif"), Array("Basique", nB),
Array("Opérationnel™, n0O), Array("Avancé", nA))
End Sub

Etape 4 — La convocation par mail (OLE)

+ Ajoutez un bouton « Convoquer » qui, pour le dernier stagiaire saisi, prépare un mail
Outlook de convocation (méthode .Display pour vérifier avant envoi).

Sub Convoquer()

Dim ol As Object, msg As Object, ws As Worksheet, L As Long
Set ws = Worksheets("Données")
L = ws.Cells(ws.Rows.Count, 1).End(x1lUp).Row
Set ol = CreateObject("Outlook.Application™)
Set msg = ol.Createltem(9)
With msg

.Subject = "Convocation - Formation Excel VBA"

.Body = "Bonjour " & ws.Cells(L, 2) & "," & vbCrLf & _

"Votre session de formation est confirmée." & vbCrLf & _

"A3iformations™
.Display
End With
End Sub
Astuce

Reliez chaque macro a un bouton de la feuille ou a une icéne du ruban : I'utilisateur final n’a alors
jamais a ouvrir I'éditeur VBE.

Pour aller plus loin : ajoutez une gestion d’erreurs globale, protégez le projet par mot de
passe, et transformez les macros communes en complément (.xlam) réutilisable sur tous vos
classeurs.

Annexe A — Raccourcis essentiels de I'éditeur VBE

Raccourci | Action

Alt + F11 Basculer entre Excel et I'éditeur VBE
Alt + F8 Ouvrir la liste des macros (depuis Excel)
F5 Exécuter la procédure courante

F8 Exécuter en pas a pas détaillé

Maj + F8 Pas a pas principal (sans entrer dans les sous-procédures)
F9 Activer / désactiver un point d’arrét

Ctrl + G Afficher la fenétre Exécution immédiate
F2 Ouvrir 'Explorateur d’objets

F4 Afficher la fenétre Propriétés

Ctrl + Espace Compléter automatiquement un mot-clé
Echap / Ctrl + Pause Interrompre une macro en cours

Annexe B — Glossaire

Terme Définition

Macro Suite d’instructions automatisant une tache.

VBE Visual Basic Editor : I'éditeur ou I'on écrit le code.

Module Conteneur de code (standard, d’objet ou de classe).
Procédure Sub (actions) ou Function (renvoie une valeur).

Variable Nom désignant une valeur stockée en mémoire.

Objet Elément manipulable (Range, Worksheet...) avec propriétés et méthodes.
Propriété Caractéristique d’'un objet (Value, Font...).

Méthode Action qu’un objet sait exécuter (Copy, Select...).

Evénement Procédure déclenchée automatiquement (Workbook Open...).
UserForm Formulaire personnalisé regroupant des contréles.

Argument Valeur transmise a une procédure ou fonction.

Débogage Recherche et correction des erreurs d’un code.

Annexe C — Erreurs fréquentes et solutions

Symptéme Cause probable

Le code disparait a la Classeur enregistré en .xlIsx

fermeture

« Variable non définie » Option Explicit + faute de
frappe

« Objet requis » Set oublié sur une variable objet

Résultat faux et silencieux Variable mal orthographiée

InputBox renvoie une erreur Texte non converti en nombre

Macro qui ne s’arréte plus Boucle infinie

« Macros désactivées » Sécurité des macros

Solution

Enregistrer en .xIsm

Corriger ou déclarer la variable

Set obj = ..

Activer Option Explicit
CDbl / vérifier le vide
Echap, revoir la condition

Centre de gestion de la
confidentialité

Annexe D — Grille d’auto-évaluation

Cochez les compétences que vous estimez maitriser a I'issue de la formation. Cette grille vous
aide a cibler les points a retravaliller.

Compétence Acquis A revoir
Afficher 'onglet Développeur et ouvrir le VBE
Enregistrer, lancer et tracer une macro (F8)

Utiliser les fenétres de débogage

Déclarer des variables typées et des constantes
Manipuler I'objet Range (Set, With, Offset)

Créer une énumération et un graphique par le code
Utiliser MsgBox et InputBox

Concevoir un UserForm et ses controles

Ecrire des Sub et des Function avec arguments
Maitriser conditions, boucles et opérateurs

Ecrire une procédure d’événement

Gérer les erreurs (On Error)

— Fin du support Excel VBA —

